

# A Computational Reconstruction of Homotopy Type Theory for Finite Types

## Abstract

Homotopy type theory (HoTT) relates some aspects of topology, algebra, geometry, physics, logic, and type theory, in a unique novel way that promises a new and foundational perspective on mathematics and computation. The heart of HoTT is the *univalence axiom*, which informally states that isomorphic structures can be identified. One of the major open problems in HoTT is a computational interpretation of this axiom. We propose that, at least for the special case of finite types, reversible computation via type isomorphisms is the computational interpretation of univalence.

## 1. Introduction

**Conventional HoTT/Agda approach** We start with a computational framework: data (pairs, etc.) and functions between them. There are computational rules (beta, etc.) that explain what a function does on a given datum.

We then have a notion of identity which we view as a process that equates two things and model as a new kind of data. Initially we only have identities between beta-equivalent things.

Then we postulate a process that identifies any two functions that are extensionally equivalent. We also postulate another process that identifies any two sets that are isomorphic. This is done by adding new kinds of data for these kinds of identities.

**Our approach** Our approach is to start with a computational framework that has finite data and permutations as the operations between them. The computational rules apply permutations.

HoTT [The Univalent Foundations Program 2013] says id types are an inductively defined type family with `refl` as constructor. We say it is a family defined with `pi` combinators as constructors. Replace path induction with `refl` as base case with our induction.

**Generalization** How would that generalize to first-class functions? Using negative and fractionals? Groupoids?

In a computational world in which the laws of physics are embraced and resources are carefully maintained (e.g., quantum computing [Abramsky and Coecke 2004; Nielsen and Chuang 2000]), programs must be reversible. Although this is apparently a limiting idea, it turns out that conventional computation can be viewed as a special case of such resource-preserving reversible programs. This thesis has been explored for many years from different perspectives [Bennett 2003, 2010, 1973; Fredkin and Toffoli 1982; Landauer 1961, 1996; Toffoli 1980]. We build on the work of James

and Sabry [2012a] which expresses this thesis in a type theoretic computational framework, expressing computation via type isomorphisms.

## 2. Condensed Background on HoTT

Informally, and as a first approximation, one may think of HoTT as a variation on Martin-Löf type theory in which all equalities are given *computational content*. We explain the basic ideas below.

### 2.1 Paths

Formally, Martin-Löf type theory, is based on the principle that every proposition, i.e., every statement that is susceptible to proof, can be viewed as a type. Indeed, if a proposition  $P$  is true, the corresponding type is inhabited and it is possible to provide evidence or proof for  $P$  using one of the elements of the type  $P$ . If, however, a proposition  $P$  is false, the corresponding type is empty and it is impossible to provide a proof for  $P$ . The type theory is rich enough to express the standard logical propositions denoting conjunction, disjunction, implication, and existential and universal quantifications. In addition, it is clear that the question of whether two elements of a type are equal is a proposition, and hence that this proposition must correspond to a type. In Agda, one may write proofs of these propositions as shown in the two examples below:

```
i0 : 3 ≡ 3
i0 = refl 3


i1 : (1 + 2) ≡ (3 * 1)
i1 = refl 3
```

More generally, given two values  $m$  and  $n$  of type  $\mathbb{N}$ , it is possible to construct an element `refl`  $k$  of the type  $m \equiv n$  if and only if  $m$ ,  $n$ , and  $k$  are all “equal.” As shown in example `i1`, this notion of *propositional equality* is not just syntactic equality but generalizes to *definitional equality*, i.e., to equality that can be established by normalizing the two values to their normal forms.


The important question from the HoTT perspective is the following: given two elements  $p$  and  $q$  of some type  $x \equiv y$  with  $x, y : A$ , what can we say about the elements of type  $p \equiv q$ . Or, in more familiar terms, given two proofs of some proposition  $P$ , are these two proofs themselves “equal.” In some situations, the only interesting property of proofs is their existence, i.e., all proofs of the same proposition are considered equivalent. A twist that dates back to a paper by Hofmann and Streicher [1996] is that proofs actually possess a structure of great combinatorial complexity. HoTT builds on this idea by interpreting types as topological spaces or weak  $\infty$ -groupoids, and interpreting identities between elements of a type  $x \equiv y$  as *paths* from the point  $x$  to the point  $y$ . If  $x$  and  $y$  are themselves paths, the elements of  $x \equiv y$  become paths between paths, or homotopies in the topological language. To be explicit, we will often refer to types as *spaces* which consist of *points*, paths, 2-paths, etc. and write  $\equiv_A$  for the type of paths in space  $A$ .

As a simple example, we are used to thinking of types as sets of values. So we typically view the type `Bool` as the figure on the left

but in HoTT we should instead think about it as the figure on the right where there is a (trivial) path `refl b` from each point `b` to itself:



In this particular case, it makes no difference, but in general we may have a much more complicated path structure. The classical such example is the topological *circle* which is a space consisting of a point `base` and a *non trivial* path `loop` from `base` to itself. As stated, this does not amount to much. However, because paths carry additional structure (explained below), that space has the following non-trivial structure:



The additional structure of types is formalized as follows. Let  $x$ ,  $y$ , and  $z$  be elements of some space  $A$ :

- For every path  $p : x \equiv_A y$ , there exists a path  $!p : y \equiv_A x$ ;
- For every pair of paths  $p : x \equiv_A y$  and  $q : y \equiv_A z$ , there exists a path  $p \odot q : x \equiv_A z$ ;
- Subject to the following conditions:
  - $p \odot \text{refl } y \equiv_{(x \equiv_A y)} p$ ;
  - $p \equiv_{(x \equiv_A y)} \text{refl } x \odot p$
  - $!p \odot p \equiv_{(y \equiv_A y)} \text{refl } y$
  - $p \odot !p \equiv_{(x \equiv_A x)} \text{refl } x$
  - $!(!p) \equiv_{(x \equiv_A y)} p$
  - $p \odot (q \odot r) \equiv_{(x \equiv_A z)} (p \odot q) \odot r$
- This structure repeats one level up and so on ad infinitum.

## 2.2 Univalence

In addition to paths between the points `false` and `true` in the space `Bool`, it is also possible to consider paths between the space `Bool` and itself by considering `Bool` as a “point” in the universe `Set` of types. As usual, we have the trivial path which is given by the constructor `refl`:

```
p : Bool ≡ Bool
p = refl Bool
```

There are, however, other (non trivial) paths between `Bool` and itself and they are justified by the *univalence* axiom. As an example, the remainder of this section justifies that there is a path between `Bool` and itself corresponding to the boolean negation function.

We begin by formalizing the equivalence of functions  $\sim$ . Intuitively, two functions are equivalent if their results are propositionally equal for all inputs. A function  $f : A \rightarrow B$  is called an *equivalence* if there are functions  $g$  and  $h$  with whom its composition is the identity. Finally two spaces  $A$  and  $B$  are equivalent,  $A \simeq B$ , if there is an equivalence between them:

```
≃ _ ≃ _ : ∀ {ℓ ℓ'} → {A : Set ℓ} {B : Set ℓ'} →
  (f g : (x : A) → P x) → Set (ℓ ⊔ ℓ')
≃ _ ≃ _ {ℓ} {ℓ'} {A} {B} f g = (x : A) → f x ≡ g x
```

```
record isequiv {ℓ ℓ'} {A : Set ℓ} {B : Set ℓ'} (f : A → B)
  : Set (ℓ ⊔ ℓ') where
  constructor mkisequiv
  field
    g : B → A
    α : (f ∘ g) ~ id
    h : B → A
    β : (h ∘ f) ~ id
```

```
≃ _ ≃ _ : ∀ {ℓ ℓ'} (A : Set ℓ) (B : Set ℓ') → Set (ℓ ⊔ ℓ')
A ≃ B = Σ (A → B) isequiv
```

We can now formally state the univalence axiom:

```
postulate univalence : {A B : Set} → (A ≡ B) ≃ (A ≃ B)
```

For our purposes, the important consequence of the univalence axiom is that equivalence of spaces implies the existence of a path between the spaces. In other words, in order to assert the existence of a path *notpath* between `Bool` and itself, we need to prove that the boolean negation function is an equivalence between the space `Bool` and itself, as shown below:

```
not2~id : (not ∘ not) ~ id
not2~id false = refl false
not2~id true  = refl true
```

```
notequiv : Bool ≃ Bool
notequiv = (not ,
  record {
    g = not ; α = not2~id ; h = not ; β = not2~id
  })
```

```
notpath h : Bool ≡ Bool
notpath with univalence
... | ( , eq) = isequiv.g eq notequiv
```

Although the code asserting the existence of a non trivial path between `Bool` and itself “compiles,” it is no longer executable as it relies on an Agda postulate. We analyze the situation from the perspective of reversible programming languages based on type isomorphisms [Bowman et al. 2011; James and Sabry 2012a,b].

## 3. Computing with Type Isomorphisms

The main syntactic vehicle for the technical developments in this paper is a simple language called  $\Pi$  whose only computations are isomorphisms between finite types [2012a]. After reviewing the motivation for this language and its relevance to HoTT, we present its syntax and semantics.

### 3.1 Reversibility

In a computational world in which the laws of physics are embraced and resources are carefully maintained (e.g., quantum computing [Abramsky and Coecke 2004; Nielsen and Chuang 2000]), programs must be reversible. Although this is apparently a limiting idea, it turns out that conventional computation can be viewed as a special case of such resource-preserving reversible programs. This thesis has been explored for many years from different perspectives [Bennett 2003, 2010, 1973; Fredkin and Toffoli 1982; Landauer 1961, 1996; Toffoli 1980].

The relevance of reversibility to HoTT is based on the following analysis. The conventional HoTT approach starts with two, a priori, different notions: functions and paths, and then postulates an equivalence between a particular class of functions and paths. As illustrated above, some functions like *not* correspond to paths. Most functions, however, are evidently unrelated to paths. In particular, any function of type  $A \rightarrow B$  that does not have an inverse of type  $B \rightarrow A$  cannot have any direct correspondence to paths as

all paths have inverses. An interesting question then poses itself: since reversible computational models — in which all functions have inverses — are known to be universal computational models, what would happen if we considered a variant of HoTT based exclusively on reversible functions? Presumably in such a variant, all functions — being reversible — would potentially correspond to paths and the distinction between the two notions would vanish making the univalence postulate unnecessary. This is the precise idea we investigate in detail in the remainder of the paper.

### 3.2 Syntax and Semantics of $\Pi$

The  $\Pi$  family of languages is based on type isomorphisms. In the variant we consider, the set of types  $\tau$  includes the empty type 0, the unit type 1, and conventional sum and product types. The values classified by these types are the conventional ones:  $()$  of type 1,  $\text{inl } v$  and  $\text{inr } v$  for injections into sum types, and  $(v_1, v_2)$  for product types:

(Types)	$\tau ::= 0 \mid 1 \mid \tau_1 + \tau_2 \mid \tau_1 * \tau_2$
(Values)	$v ::= () \mid \text{inl } v \mid \text{inr } v \mid (v_1, v_2)$
(Combinator types)	$\tau_1 \leftrightarrow \tau_2$
(Combinators)	$c ::= [\text{see Table 1}]$

The interesting syntactic category of  $\Pi$  is that of *combinators* which are witnesses for type isomorphisms  $\tau_1 \leftrightarrow \tau_2$ . They consist of base combinators (on the left side of Table 1) and compositions (on the right side of the same table). Each line of the table on the left introduces a pair of dual constants<sup>1</sup> that witness the type isomorphism in the middle. This set of isomorphisms is known to be complete [Fiore 2004; Fiore et al. 2006] and the language is universal for hardware combinational circuits [James and Sabry 2012a].<sup>2</sup>

From the perspective of category theory, the language  $\Pi$  models what is called a *symmetric bimonoidal category* or a *commutative rig category*. These are categories with two binary operations  $\oplus$  and  $\otimes$  satisfying the axioms of a rig (i.e., a ring without negative elements also known as a semiring) up to coherent isomorphisms. And indeed the types of the  $\Pi$ -combinators are precisely the semiring axioms. A formal way of saying this is that  $\Pi$  is the *categorification* [Baez and Dolan 1998] of the natural numbers. A simple (slightly degenerate) example of such categories is the category of finite sets and permutations in which we interpret every  $\Pi$ -type as a finite set, the values as elements in these finite sets, and the combinators as permutations. In the remainder of this paper, we will more interested in a model based on groupoids. But first, we give an operational semantics for  $\Pi$ .

Operationally, the semantics consists of a pair of mutually recursive evaluators that take a combinator and a value and propagate the value in the “forward”  $\triangleright$  direction or in the “backwards”  $\triangleleft$  direction. We show the complete forward evaluator; the backwards

evaluator differs in trivial ways:

$\text{identl}_+ \triangleright (\text{inr } v)$	$= v$
$\text{identr}_+ \triangleright v$	$= \text{inr } v$
$\text{swap}_+ \triangleright (\text{inl } v)$	$= \text{inr } v$
$\text{swap}_+ \triangleright (\text{inr } v)$	$= \text{inl } v$
$\text{assocl}_+ \triangleright (\text{inl } v)$	$= \text{inl } (\text{inl } v)$
$\text{assocl}_+ \triangleright (\text{inr } (\text{inl } v))$	$= \text{inl } (\text{inr } v)$
$\text{assocl}_+ \triangleright (\text{inr } (\text{inr } v))$	$= \text{inr } v$
$\text{assocr}_+ \triangleright (\text{inl } (\text{inl } v))$	$= \text{inl } v$
$\text{assocr}_+ \triangleright (\text{inl } (\text{inr } v))$	$= \text{inr } (\text{inl } v)$
$\text{assocr}_+ \triangleright (\text{inr } v)$	$= \text{inr } (\text{inr } v)$
$\text{identl}_* \triangleright ((), v)$	$= v$
$\text{identr}_* \triangleright v$	$= ((), v)$
$\text{swap}_* \triangleright (v_1, v_2)$	$= (v_2, v_1)$
$\text{assocl}_* \triangleright (v_1, (v_2, v_3))$	$= ((v_1, v_2), v_3)$
$\text{assocr}_* \triangleright ((v_1, v_2), v_3)$	$= (v_1, (v_2, v_3))$
$\text{dist} \triangleright (\text{inl } v_1, v_3)$	$= \text{inl } (v_1, v_3)$
$\text{dist} \triangleright (\text{inr } v_2, v_3)$	$= \text{inr } (v_2, v_3)$
$\text{factor} \triangleright (\text{inl } (v_1, v_3))$	$= (\text{inl } v_1, v_3)$
$\text{factor} \triangleright (\text{inr } (v_2, v_3))$	$= (\text{inr } v_2, v_3)$
$\text{id} \triangleright v$	$= v$
$(\text{sym } c) \triangleright v$	$= c \triangleleft v$
$(c_1 \circ c_2) \triangleright v$	$= c_2 \triangleright (c_1 \triangleright v)$
$(c_1 \oplus c_2) \triangleright (\text{inl } v)$	$= \text{inl } (c_1 \triangleright v)$
$(c_1 \oplus c_2) \triangleright (\text{inr } v)$	$= \text{inr } (c_2 \triangleright v)$
$(c_1 \otimes c_2) \triangleright (v_1, v_2)$	$= (c_1 \triangleright v_1, c_2 \triangleright v_2)$

### 3.3 Groupoid Model

Instead of modeling the types of  $\Pi$  using sets and the combinators using permutations on the sets, we use a semantics that identifies  $\Pi$  combinators with *paths*. More precisely, we model the universe of  $\Pi$  types as a space  $U$  whose points are the individual  $\Pi$ -types (which are themselves spaces  $t$  containing points). We then postulate that there is path between the spaces  $t_1$  and  $t_2$  if there is a  $\Pi$  combinator  $c : t_1 \leftrightarrow t_2$ . Our postulate is similar in spirit to the univalence axiom but, unlike the latter, it has a simple computational interpretation. A path directly corresponds to a type isomorphism with a clear operational semantics as presented in the previous section. As we will explain in more detail below, this approach replaces the datatype  $\equiv$  modeling propositional equality with the datatype  $\leftrightarrow$  modeling type isomorphisms. With this switch, the  $\Pi$ -combinators of Table 1 become *syntax* for the paths in the space  $U$ . Put differently, instead of having exactly one constructor `refl` for paths with all other paths discovered by proofs (see Secs. 2.5–2.12 of the HoTT book [?]) or postulated by the univalence axiom, we have an *inductive definition* that completely specifies all the paths in the space  $U$ .

We begin with the datatype definition of the universe  $U$  of finite types which are constructed using **ZERO**, **ONE**, **PLUS**, and **TIMES**. Each of these finite types will correspond to a set of points with paths connecting some of the points. The underlying set is computed by  $\llbracket \_ \rrbracket$  as follows: **ZERO** maps to the empty set  $\perp$ , **ONE** maps to the singleton set  $\top$ , **PLUS** maps to the disjoint union  $\uplus$ , and **TIMES** maps to the cartesian product  $\times$ .

```
data U : Set where
  ZERO  : U
  ONE   : U
  PLUS  : U → U → U
  TIMES : U → U → U
```

```
 $\llbracket \_ \rrbracket : U \rightarrow \text{Set}$ 
 $\llbracket \text{ZERO} \rrbracket = \perp$ 
 $\llbracket \text{ONE} \rrbracket = \top$ 
 $\llbracket \text{PLUS } t_1 \ t_2 \rrbracket = \llbracket t_1 \rrbracket \uplus \llbracket t_2 \rrbracket$ 
 $\llbracket \text{TIMES } t_1 \ t_2 \rrbracket = \llbracket t_1 \rrbracket \times \llbracket t_2 \rrbracket$ 
```

<sup>1</sup> where  $\text{swap}_+$  and  $\text{swap}_*$  are self-dual.

<sup>2</sup> If recursive types and a trace operator are added, the language becomes Turing complete [Bowman et al. 2011; James and Sabry 2012a]. We will not be concerned with this extension in the main body of this paper but it will be briefly discussed in the conclusion.

$identl_+$	$0 + \tau \leftrightarrow \tau$	$: identr_+$
$swap_+$	$\tau_1 + \tau_2 \leftrightarrow \tau_2 + \tau_1$	$: swap_+$
$assocl_+$	$\tau_1 + (\tau_2 + \tau_3) \leftrightarrow (\tau_1 + \tau_2) + \tau_3$	$: assocr_+$
$identl_*$	$1 * \tau \leftrightarrow \tau$	$: identr_*$
$swap_*$	$\tau_1 * \tau_2 \leftrightarrow \tau_2 * \tau_1$	$: swap_*$
$assocl_*$	$\tau_1 * (\tau_2 * \tau_3) \leftrightarrow (\tau_1 * \tau_2) * \tau_3$	$: assocr_*$
$dist_0$	$0 * \tau \leftrightarrow 0$	$: factor_0$
$dist$	$(\tau_1 + \tau_2) * \tau_3 \leftrightarrow (\tau_1 * \tau_3) + (\tau_2 * \tau_3)$	$: factor$

$\frac{}{\vdash id : \tau \leftrightarrow \tau}$	$\frac{\vdash c : \tau_1 \leftrightarrow \tau_2}{\vdash sym\ c : \tau_2 \leftrightarrow \tau_1}$
$\frac{}{\vdash c_1 : \tau_1 \leftrightarrow \tau_2}$	$\frac{}{\vdash c_2 : \tau_2 \leftrightarrow \tau_3}$
$\vdash c_1 \circ c_2 : \tau_1 \leftrightarrow \tau_3$	
$\vdash c_1 : \tau_1 \leftrightarrow \tau_2$	$\vdash c_2 : \tau_2 \leftrightarrow \tau_4$
$\vdash c_1 \oplus c_2 : \tau_1 + \tau_3 \leftrightarrow \tau_2 + \tau_4$	
$\vdash c_1 : \tau_1 \leftrightarrow \tau_2$	$\vdash c_2 : \tau_3 \leftrightarrow \tau_4$
$\vdash c_1 \otimes c_2 : \tau_1 * \tau_3 \leftrightarrow \tau_2 * \tau_4$	

**Table 1.**  $\Pi$ -combinators [James and Sabry 2012a]

Paths are ultimately defined between points: in order to be able to identify them with  $\Pi$  combinators, we refine the latter to operate on *pointed sets* (or referred to as pointed types or pointed spaces) instead of the unpointed sets above. A pointed set  $\bullet[ \mid t \mid , v ]$  is a set  $t : U$  with a distinguished value  $v : \llbracket t \rrbracket$ :

```
record U• : Set where
  constructor •[_ , _]
  field
    |_| : U
    • : [ |_| ]
```

The refinement of the  $\Pi$  combinators to combinators on pointed spaces is given by the inductive family in Table ???. The definition effectively folds the operational semantics of each combinator into a path that connects its input to its output. The definition also evidently generalizes the usual propositional equality into a *heterogeneous* equality that connects points that may be in different spaces. What used to be the only constructor for paths `refl` is now just one of the many constructors (named `id↔` in the Table). Among the new constructors, we have `sym↔` that constructs path inverses, `⊙` that constructs path compositions, and `swap1+` and `swap2+` that are essentially the encoding of the path *notpath* from the space *Bool* to itself. To see this, note that *Bool* can be viewed as a shorthand for **PLUS ONE ONE** with `true` and `false` as shorthands for `inj1 tt` and `inj2 tt`. With this in mind, the path corresponding to boolean negation consists of two “fibers.”

```
BOOL : U
BOOL = PLUS ONE ONE
```

```
TRUE FALSE : [ BOOL ]
TRUE      = inj1 tt
FALSE     = inj2 tt
```

```
notpath1 : •[ BOOL , TRUE ] ↔ •[ BOOL , FALSE ]
notpath1 = swap1+
```

```
notpath2 : •[ BOOL , FALSE ] ↔ •[ BOOL , TRUE ]
notpath2 = swap2+
```

## 4. Computing with Paths

The evaluation of a program is not done in order to figure out the output value. Both the input and output values are encoded in the type of the program; what the evaluation does is follow the path to constructively reach the output value from the input value. Even though programs of the same pointed types are, by definition, observationally equivalent, they may follow different paths. At this point, we simply declare that all such programs are “the same.” At the next level, we will weaken this “path irrelevant” equivalence and reason about which paths can be equated to other paths via 2paths etc.

Simplify various compositions

```
simplifySym : {t1 t2 : U•} → (c1 : t1 ↔ t2) → (t2 ↔ t1)
simplifySym unite+ = uniti+
simplifySym uniti+ = unite+
simplifySym swap1+ = swap2+
simplifySym swap2+ = swap1+
simplifySym assoc1+ = assocr1+
simplifySym assoc2+ = assocr2+
simplifySym assoc3+ = assocr3+
simplifySym assocr1+ = assocl1+
simplifySym assocr2+ = assocl2+
simplifySym assocr3+ = assocl3+
simplifySym unite* = uniti*
simplifySym uniti* = unite*
simplifySym swap* = swap*
simplifySym assocl* = assocr*
simplifySym assocr* = assocl*
simplifySym distz = factorz
simplifySym factorz = distz
simplifySym dist1 = factor1
simplifySym dist2 = factor2
simplifySym factor1 = dist1
simplifySym factor2 = dist2
simplifySym id↔ = id↔
simplifySym (sym↔ c) = c
simplifySym (c1 ⊙ c2) = simplifySym c2 ⊙ simplifySym c1
simplifySym (c1 ⊕1 c2) = simplifySym c1 ⊕1 simplifySym c2
simplifySym (c1 ⊕2 c2) = simplifySym c1 ⊕2 simplifySym c2
simplifySym (c1 ⊗ c2) = simplifySym c1 ⊗ simplifySym c2
```

```
simplify|⊙ : {t1 t2 t3 : U•} →
  (c1 : t1 ↔ t2) → (c2 : t2 ↔ t3) → (t1 ↔ t3)
simplify|⊙ id↔ c = c
simplify|⊙ unite+ uniti+ = id↔
simplify|⊙ uniti+ unite+ = id↔
simplify|⊙ swap1+ swap2+ = id↔
simplify|⊙ swap2+ swap1+ = id↔
simplify|⊙ assocl1+ assocr1+ = id↔
simplify|⊙ assocl2+ assocr2+ = id↔
simplify|⊙ assocl3+ assocr3+ = id↔
simplify|⊙ assocr1+ assocl1+ = id↔
simplify|⊙ assocr2+ assocl2+ = id↔
simplify|⊙ assocr3+ assocl3+ = id↔
simplify|⊙ unite* uniti* = id↔
simplify|⊙ uniti* unite* = id↔
simplify|⊙ swap* swap* = id↔
simplify|⊙ assocl* assocr* = id↔
simplify|⊙ assocr* assocl* = id↔
simplify|⊙ factorz distz = id↔
simplify|⊙ dist1 factor1 = id↔
simplify|⊙ dist2 factor2 = id↔
simplify|⊙ factor1 dist1 = id↔
simplify|⊙ factor2 dist2 = id↔
simplify|⊙ (c1 ⊙ c2) c3 = c1 ⊙ (c2 ⊙ c3)
simplify|⊙ (c1 ⊕1 c2) swap1+ = swap1+ ⊙ (c2 ⊕2 c1)
```

data  $\_ \leftrightarrow \_ : \mathbf{U} \bullet \rightarrow \mathbf{U} \bullet \rightarrow \text{Set where}$

$\text{unite}_+ : \forall \{t\ v\} \rightarrow \bullet[\text{PLUS ZERO } t, \text{inj}_2\ v] \leftrightarrow \bullet[t, v]$   
 $\text{unit}_+ : \forall \{t\ v\} \rightarrow \bullet[t, v] \leftrightarrow \bullet[\text{PLUS ZERO } t, \text{inj}_2\ v]$   
 $\text{swap1}_+ : \forall \{t_1\ t_2\ v_1\} \rightarrow$   
 $\quad \bullet[\text{PLUS } t_1\ t_2, \text{inj}_1\ v_1] \leftrightarrow \bullet[\text{PLUS } t_2\ t_1, \text{inj}_2\ v_1]$   
 $\text{swap2}_+ : \forall \{t_1\ t_2\ v_2\} \rightarrow$   
 $\quad \bullet[\text{PLUS } t_1\ t_2, \text{inj}_2\ v_2] \leftrightarrow \bullet[\text{PLUS } t_2\ t_1, \text{inj}_1\ v_2]$   
 $\text{assoc1}_+ : \forall \{t_1\ t_2\ t_3\ v_1\} \rightarrow$   
 $\quad \bullet[\text{PLUS } t_1 (\text{PLUS } t_2\ t_3), \text{inj}_1\ v_1] \leftrightarrow$   
 $\quad \bullet[\text{PLUS } (\text{PLUS } t_1\ t_2)\ t_3, \text{inj}_1 (\text{inj}_1\ v_1)]$   
 $\text{assoc2}_+ : \forall \{t_1\ t_2\ t_3\ v_2\} \rightarrow$   
 $\quad \bullet[\text{PLUS } t_1 (\text{PLUS } t_2\ t_3), \text{inj}_2 (\text{inj}_1\ v_2)] \leftrightarrow$   
 $\quad \bullet[\text{PLUS } (\text{PLUS } t_1\ t_2)\ t_3, \text{inj}_2 (\text{inj}_2\ v_2)]$   
 $\text{assoc3}_+ : \forall \{t_1\ t_2\ t_3\ v_3\} \rightarrow$   
 $\quad \bullet[\text{PLUS } t_1 (\text{PLUS } t_2\ t_3), \text{inj}_2 (\text{inj}_2\ v_3)] \leftrightarrow$   
 $\quad \bullet[\text{PLUS } (\text{PLUS } t_1\ t_2)\ t_3, \text{inj}_2\ v_3]$   
 $\text{assocr1}_+ : \forall \{t_1\ t_2\ t_3\ v_1\} \rightarrow$   
 $\quad \bullet[\text{PLUS } (\text{PLUS } t_1\ t_2)\ t_3, \text{inj}_1 (\text{inj}_1\ v_1)] \leftrightarrow$   
 $\quad \bullet[\text{PLUS } t_1 (\text{PLUS } t_2\ t_3), \text{inj}_1\ v_1]$   
 $\text{assocr2}_+ : \forall \{t_1\ t_2\ t_3\ v_2\} \rightarrow$   
 $\quad \bullet[\text{PLUS } (\text{PLUS } t_1\ t_2)\ t_3, \text{inj}_1 (\text{inj}_2\ v_2)] \leftrightarrow$   
 $\quad \bullet[\text{PLUS } t_1 (\text{PLUS } t_2\ t_3), \text{inj}_2 (\text{inj}_1\ v_2)]$   
 $\text{assocr3}_+ : \forall \{t_1\ t_2\ t_3\ v_3\} \rightarrow$   
 $\quad \bullet[\text{PLUS } (\text{PLUS } t_1\ t_2)\ t_3, \text{inj}_2\ v_3] \leftrightarrow$   
 $\quad \bullet[\text{PLUS } t_1 (\text{PLUS } t_2\ t_3), \text{inj}_2 (\text{inj}_2\ v_3)]$   
 $\text{unite}_* : \forall \{t\ v\} \rightarrow \bullet[\text{TIMES ONE } t, (\text{tt}, v)] \leftrightarrow \bullet[t, v]$   
 $\text{unit}_* : \forall \{t\ v\} \rightarrow \bullet[t, v] \leftrightarrow \bullet[\text{TIMES ONE } t, (\text{tt}, v)]$   
 $\text{swap}_* : \forall \{t_1\ t_2\ v_1\ v_2\} \rightarrow$   
 $\quad \bullet[\text{TIMES } t_1\ t_2, (v_1, v_2)] \leftrightarrow \bullet[\text{TIMES } t_2\ t_1, (v_2, v_1)]$   
 $\text{assoc}_* : \forall \{t_1\ t_2\ t_3\ v_1\ v_2\ v_3\} \rightarrow$   
 $\quad \bullet[\text{TIMES } t_1 (\text{TIMES } t_2\ t_3), (v_1, (v_2, v_3))] \leftrightarrow$   
 $\quad \bullet[\text{TIMES } (\text{TIMES } t_1\ t_2)\ t_3, ((v_1, v_2), v_3)]$   
 $\text{assocr}_* : \forall \{t_1\ t_2\ t_3\ v_1\ v_2\ v_3\} \rightarrow$

$\bullet[\text{TIMES } (\text{TIMES } t_1\ t_2)\ t_3, ((v_1, v_2), v_3)] \leftrightarrow$   
 $\bullet[\text{TIMES } t_1 (\text{TIMES } t_2\ t_3), (v_1, (v_2, v_3))]$   
 $\text{distz} : \forall \{t\ v\ \text{absurd}\} \rightarrow$   
 $\quad \bullet[\text{TIMES ZERO } t, (\text{absurd}, v)] \leftrightarrow \bullet[\text{ZERO}, \text{absurd}]$   
 $\text{factorz} : \forall \{t\ v\ \text{absurd}\} \rightarrow$   
 $\quad \bullet[\text{ZERO}, \text{absurd}] \leftrightarrow \bullet[\text{TIMES ZERO } t, (\text{absurd}, v)]$   
 $\text{dist1} : \forall \{t_1\ t_2\ t_3\ v_1\ v_3\} \rightarrow$   
 $\quad \bullet[\text{TIMES } (\text{PLUS } t_1\ t_2)\ t_3, (\text{inj}_1\ v_1, v_3)] \leftrightarrow$   
 $\quad \bullet[\text{PLUS } (\text{TIMES } t_1\ t_3) (\text{TIMES } t_2\ t_3), \text{inj}_1 (v_1, v_3)]$   
 $\text{dist2} : \forall \{t_1\ t_2\ t_3\ v_2\ v_3\} \rightarrow$   
 $\quad \bullet[\text{TIMES } (\text{PLUS } t_1\ t_2)\ t_3, (\text{inj}_2\ v_2, v_3)] \leftrightarrow$   
 $\quad \bullet[\text{PLUS } (\text{TIMES } t_1\ t_3) (\text{TIMES } t_2\ t_3), \text{inj}_2 (v_2, v_3)]$   
 $\text{factor1} : \forall \{t_1\ t_2\ t_3\ v_1\ v_3\} \rightarrow$   
 $\quad \bullet[\text{PLUS } (\text{TIMES } t_1\ t_3) (\text{TIMES } t_2\ t_3), \text{inj}_1 (v_1, v_3)] \leftrightarrow$   
 $\quad \bullet[\text{TIMES } (\text{PLUS } t_1\ t_2)\ t_3, (\text{inj}_1\ v_1, v_3)]$   
 $\text{factor2} : \forall \{t_1\ t_2\ t_3\ v_2\ v_3\} \rightarrow$   
 $\quad \bullet[\text{PLUS } (\text{TIMES } t_1\ t_3) (\text{TIMES } t_2\ t_3), \text{inj}_2 (v_2, v_3)] \leftrightarrow$   
 $\quad \bullet[\text{TIMES } (\text{PLUS } t_1\ t_2)\ t_3, (\text{inj}_2\ v_2, v_3)]$   
 $\text{id} \leftrightarrow : \forall \{t\ v\} \rightarrow \bullet[t, v] \leftrightarrow \bullet[t, v]$   
 $\text{sym} \leftrightarrow : \forall \{t_1\ t_2\ v_1\ v_2\} \rightarrow (\bullet[t_1, v_1] \leftrightarrow \bullet[t_2, v_2]) \rightarrow$   
 $\quad (\bullet[t_2, v_2] \leftrightarrow \bullet[t_1, v_1])$   
 $\_ \odot \_ : \forall \{t_1\ t_2\ t_3\ v_1\ v_2\ v_3\} \rightarrow (\bullet[t_1, v_1] \leftrightarrow \bullet[t_2, v_2]) \rightarrow$   
 $\quad (\bullet[t_2, v_2] \leftrightarrow \bullet[t_3, v_3]) \rightarrow (\bullet[t_1, v_1] \leftrightarrow \bullet[t_3, v_3])$   
 $\_ \oplus 1 \_ : \forall \{t_1\ t_2\ t_3\ t_4\ v_1\ v_2\ v_3\ v_4\} \rightarrow$   
 $\quad (\bullet[t_1, v_1] \leftrightarrow \bullet[t_3, v_3]) \rightarrow (\bullet[t_2, v_2] \leftrightarrow \bullet[t_4, v_4]) \rightarrow$   
 $\quad (\bullet[\text{PLUS } t_1\ t_2, \text{inj}_1\ v_1] \leftrightarrow \bullet[\text{PLUS } t_3\ t_4, \text{inj}_1\ v_3])$   
 $\_ \oplus 2 \_ : \forall \{t_1\ t_2\ t_3\ t_4\ v_1\ v_2\ v_3\ v_4\} \rightarrow$   
 $\quad (\bullet[t_1, v_1] \leftrightarrow \bullet[t_3, v_3]) \rightarrow (\bullet[t_2, v_2] \leftrightarrow \bullet[t_4, v_4]) \rightarrow$   
 $\quad (\bullet[\text{PLUS } t_1\ t_2, \text{inj}_2\ v_2] \leftrightarrow \bullet[\text{PLUS } t_3\ t_4, \text{inj}_2\ v_4])$   
 $\_ \otimes \_ : \forall \{t_1\ t_2\ t_3\ t_4\ v_1\ v_2\ v_3\ v_4\} \rightarrow$   
 $\quad (\bullet[t_1, v_1] \leftrightarrow \bullet[t_3, v_3]) \rightarrow (\bullet[t_2, v_2] \leftrightarrow \bullet[t_4, v_4]) \rightarrow$   
 $\quad (\bullet[\text{TIMES } t_1\ t_2, (v_1, v_2)] \leftrightarrow \bullet[\text{TIMES } t_3\ t_4, (v_3, v_4)])$

**Table 2.** Pointed version of  $\Pi$ -combinators or inductive definition of paths

$\text{simplify} \odot (c_1 \oplus 2\ c_2) \text{ swap2}_+ = \text{swap2}_+ \odot (c_2 \oplus 1\ c_1)$   
 $\text{simplify} \odot (\_ \otimes \_ \{ \text{ONE} \} \{ \text{ONE} \} c_1\ c_2) \text{ unite}_* = \text{unite}_* \odot c_2$   
 $\text{simplify} \odot (c_1 \otimes c_2) \text{ swap}_* = \text{swap}_* \odot (c_2 \otimes c_1)$   
 $\text{simplify} \odot (c_1 \otimes c_2) (c_3 \otimes c_4) = (c_1 \odot c_3) \otimes (c_2 \odot c_4)$   
 $\text{simplify} \odot c_1\ c_2 = c_1 \odot c_2$

We need to show that the groupoid path structure is faithfully represented. The combinator  $\text{id}$  introduces all the refl  $\tau : \tau \equiv \tau$  paths in  $U$ . The adjoint  $\text{sym}$   $c$  introduces an inverse path  $!p$  for each path  $p$  introduced by  $c$ . The composition operator  $\circ$  introduces a path  $p \circ q$  for every pair of paths whose endpoints match. In addition, we get paths like  $\text{swap}_+$  between  $\tau_1 + \tau_2$  and  $\tau_2 + \tau_1$ . The existence of such paths in the conventional HoTT needs to be proved from first principles for some types and *postulated* for the universe type by the univalence axiom. The  $\otimes$ -composition gives a path  $(p, q) : (\tau_1 * \tau_2) \equiv (\tau_3 * \tau_4)$  whenever we have paths  $p : \tau_1 \equiv \tau_3$  and  $q : \tau_2 \equiv \tau_4$ . A similar situation for the  $\oplus$ -composition. The structure of these paths must be discovered and these paths must be *proved* to exist using path induction in the conventional HoTT development. So far, this appears too good to be true, and it is. The problem is that paths in HoTT are subject to rules discussed at the end of Sec. 2. For example, it must be the case that if  $p : \tau_1 \equiv \tau_2$  that  $(p \circ \text{refl } \tau_2) \equiv_{\tau_1 \equiv \tau_2} p$ . This path lives in a higher universe: nothing in our  $\Pi$ -combinators would justify adding such a path as all our combinators map types to types. No combinator works one level up at the space of combinators and there is no such space in the first place. Clearly we are stuck unless we manage to express a notion of higher-order functions in  $\Pi$ . This would allow us to

internalize the type  $\tau_1 \leftrightarrow \tau_2$  as a  $\Pi$ -type which is then manipulated by the same combinators one level higher and so on.

Structure of Paths:

- What do paths in  $A \times B$  look like? We can prove that  $(a_1, b_1) \equiv (a_2, b_2)$  in  $A \times B$  iff  $a_1 \equiv a_2$  in  $A$  and  $b_1 \equiv b_2$  in  $B$ .
- What do paths in  $A_1 \uplus A_2$  look like? We can prove that  $\text{inj}_i\ x \equiv \text{inj}_j\ y$  in  $A_1 \uplus A_2$  iff  $i = j$  and  $x \equiv y$  in  $A_i$ .
- What do paths in  $A \rightarrow B$  look like? We cannot prove anything. Postulate function extensionality axiom.
- What do paths in  $\text{Set}_\ell$  look like? We cannot prove anything. Postulate univalence axiom.

Let's start with a few simple types built from the empty type, the unit type, sums, and products, and let's study the paths postulated by HoTT.

For every value in a type (point in a space) we have a trivial path from the value to itself:

Level 0: Types at this level are just plain sets with no interesting path structure. The path structure is defined at levels 1 and beyond.

for examples of 2 paths look at proofs of path assoc; triangle and pentagon rules

the idea I guess is that instead of having the usual evaluator where values flow, we want an evaluator that rewrites the circuit to primitive isos; for that we need some normal form for permutations and a proof that we can rewrite any circuit to this normal form

plan after that: add trace; this make obs equiv much more interesting and allows a limited form of h.o. functions via the int construction and then do the ring completion to get more complete notion of h.o. functions

Level 1: Types are sets of paths. The paths are defined at the previous level (level 0). At level 1, there is no interesting 2path structure. From the perspective of level 0, we have points with non-trivial paths between them, i.e., we have a groupoid. The paths cross type boundaries, i.e., we have heterogeneous equality

## 5. Conclusion

### References

- S. Abramsky and B. Coecke. A categorical semantics of quantum protocols. In *LICS*, 2004.
- J. C. Baez and J. Dolan. Categorification. In *Higher Category Theory*, Contemp. Math. 230, 1998, pp. 1-36., 1998.
- C. Bennett. Notes on Landauer's principle, reversible computation, and Maxwell's Demon. *Studies In History and Philosophy of Science Part B: Studies In History and Philosophy of Modern Physics*, 34(3):501-510, 2003.
- C. Bennett. Notes on the history of reversible computation. *IBM Journal of Research and Development*, 32(1):16-23, 2010.
- C. H. Bennett. Logical reversibility of computation. *IBM J. Res. Dev.*, 17: 525-532, November 1973.
- W. J. Bowman, R. P. James, and A. Sabry. Dagger Traced Symmetric Monoidal Categories and Reversible Programming. In *RC*, 2011.
- M. Fiore. Isomorphisms of generic recursive polynomial types. In *POPL*, pages 77-88. ACM, 2004.
- M. P. Fiore, R. Di Cosmo, and V. Balat. Remarks on isomorphisms in typed calculi with empty and sum types. *Annals of Pure and Applied Logic*, 141(1-2):35-50, 2006.
- E. Fredkin and T. Toffoli. Conservative logic. *International Journal of Theoretical Physics*, 21(3):219-253, 1982.
- M. Hofmann and T. Streicher. The groupoid interpretation of type theory. In *Venice Festschrift*, pages 83-111, 1996.
- R. P. James and A. Sabry. Information effects. In *POPL*, pages 73-84. ACM, 2012a.
- R. P. James and A. Sabry. Isomorphic interpreters from logically reversible abstract machines. In *RC*, 2012b.
- R. Landauer. Irreversibility and heat generation in the computing process. *IBM J. Res. Dev.*, 5:183-191, July 1961.
- R. Landauer. The physical nature of information. *Physics Letters A*, 1996.
- M. Nielsen and I. Chuang. *Quantum Computation and Quantum Information*. Cambridge University Press, Cambridge, 2000.
- The Univalent Foundations Program. *Homotopy Type Theory: Univalent Foundations of Mathematics*. <http://homotopytypetheory.org/book>, Institute for Advanced Study, 2013.
- T. Toffoli. Reversible computing. In *Proceedings of the 7th Colloquium on Automata, Languages and Programming*, pages 632-644. Springer-Verlag, 1980.